A CROSS-SECTION PLOTTING PROGRAM (CSPP) FOR GRIDDED (MAP) DATA

By Richard B. Wells, Charles R. Faust and James W. Mercer

U.S. GEOLOGICAL SURVEY
Open—File Report 76—689

(poor copy - best that could be obtained)

1976

U.S. DEPARTMENT OF THE INTERIOR
Thomas S. Kleppe, Secretary

GEOLOGICAL SURVEY
V.E. McKelvey, Director

For additional information write to:

Chief Hydrologist
U.S. Geological Survey, WRD
431 National Center
Reston, Virginia 22092

CONTENTS

FIGURES

TABLES

ABSTRACT

A FØRTRAN program that prepares the Calcomp plotter tape for drawing cross-sections from gridded map data is described and documented. The program accepts gridded data of the form Z(x,y), where Z is the value at the grid intersection, (x,y). Up to 10 sets of Z(x,y) data for a given grid may be drawn on the same section. The positions of the end points of the cross-section are arbitrary. A bilinear interpolation scheme is used to obtain the Z(x,y) values along the specified cross-section.

The program was designed to provide a visual aid for interpreting input and output data for two-dimensional (areal) finite-difference models used in aquifer analysis and geothermal reservoir simulation. The program should, however, be useful for drawing cross-sections of any multiple Z(x,y) gridded data. The computer code is called CSPP standing for Cross-Section Plotting Program.

INTRODUCTION

The purpose of this report is to describe and document a FØRTRAN computer program that draws cross-sections from gridded (map) data, using Calcomp plotter software. The program was designed to serve as a visual aid for evaluating input and output data of finite-difference models used in ground-water hydrology. The computer code is, however, general enough to be used for other applications. For example, the program could be used on gridded data supplied by the GPCP software package (California Computer Products, Inc., 1971).[1/]

Some typical hydrological applications of the plotting program are given in the example section in this report. Commonly, in ground-water simulation applications, the aquifer and its confining beds are irregular. For such problems it is helpful to draw geologic cross-sections based on the input data for the finite-difference grid (see figure 1). These cross-sections can be compared easily with available geologic sections to evaluate the accuracy of the input data (see figure 2). Output data can be evaluated in the same manner. For example, the computed values at specifiec times during a simualtion may be plotted on the same section for comparison with observed values (see figure 3).

1/ The use of brand names in this report are for identification purposes only and does not imply endorsement by the U.S. Geological Survey.



Figure 1. A rectangular grid showing the location of a section A-A'.



Figure 2. An example geologic cross-section corresponding to A-A' in figure 1.



Figure 3. An example of computed hydraulic-head profiles for various times along the section A-A' in figure 1.

PROGRAM DESCRIPTION

The object of the computer program is to communicate to the CALCOMP software the information necessary to draw the desired cross-section from given gridded data, Z(x,y). The user supplies the x- and y- coordinates of the grid lines, the z-values for each grid intersection, the x- and y- coordinates of the end points of the desired section, the number of intermediate interpolation points, and other data to control the scaling, size, and labeling of the plot. Up to 10 sets of z- values may be plotted on the same cross-section. The computer program calculates the coordinates of equally spaced intermediate points (the number of which is supplied by the user) along the section, computes the z- values at the end points and intermediate points, and makes the necessary calls to the CALCOMP software.

The numerical scheme that computes the values along the section is based on bilinear interpolation for a rectangular grid. Figure 4 shows a portion of a rectangular grid and the position of an interpolation point (x', y'). The program must first locate the grid block that contains the point (x', y'). The value, Z(x', y') may then be calculated from

$$Z(x', y') = \sum_{k=1}^{4} z_k v_k$$

where k is the number of the surrounding grid points as shown in figure 4, and z_k is the corresponding value of Z(x,y) at the grid point. The interpolating function, v_k, is obtained from (Zienkiewicz, 1971)

$$v_k = 1/4\ (1+\eta\eta_k)\ (1+\xi\xi_k)$$

where

$$\xi = 1 - 2\ \left(\frac{x_i - x}{x_i - x_{i-1}}\right)$$

$$\eta = 1 - 2\ \left(\frac{y_i - y}{y_i - y_{i-1}}\right)\ .$$



Figure 4. A portion of a typical grid showing the location of the interpolation point (x', y') and the labeling of its neighboring grid intersections.

PROGRAM USE

This program was designed to be applicable to a large variety of potential uses. The only input required is several parameters specifying dimensions, scaling, labeling of the cross-section plot, and the X,Y and Z(x,y) values (as many as ten sets of Z(x,y) are permitted).

Another input parameter allows the user to specify the amount of interpolation to be done for each curve of the cross section. The user specifies the number of equally spaced points between endpoints A and A' at which the program will compute the corresponding Z(x,y) values by interpolation. The accuracy and smoothness of each curve depends on this parameter and it should be carefully chosen.

The following control cards are necessary to run this program:

```
//JØB CARD
// EXEC FØRTCLG,REGIØN.GØ=170K
//FØRT.SYSIN DD *
        DECK
//LKED.SYSLIB DD DSN=SYS1.CALCØM1.C936.PLØTTER,DISP=SHR
//GØ.FT09F001 DD SYSØUT=(K,,<FORM#>)  [Prepares output for tape
                                       retrieval]
//GØ.SYSIN    DD *
        DATA
//
```

These control cards are used in conjunction with a Data 100 Model 78 terminal and were provided to us by Steven P. Larson, WRD. For other terminals, the normal CALCOMP software and plot-tape must be used.

The sample problem that follows used 170K bytes of core on the IBM 370/155. The core used will depend largely on the number of grid points. The program calls the following subprograms:

1) GETVAL (interpolates at intermediate points)

2) CALCOMP BASIC SOFTWARE PACKAGE (California Computer Products, Inc., 1969)

On a successful run the program will print the x,y coordinates and the Z(x,y) for each of the equally spaced interpolation points. All of the cross section points will be printed for each set of Z(x,y). If a point on the cross section lies outside the specified grid, an error message will be printed that includes the coordinates of the point, and the run will be terminated.

DATA INPUT

COLUMNS	FORMAT	NAME	DESCRIPTION
Card 1			
1-2	I2	ZS	No. of sets of data points to be input
6-15	F10.0	BEG(1)	x-coordinate of the beginning cross section point
16-25	F10.0	BEG(2)	y-coordinate of the beginning cross section point
26-35	F10.0	END(1)	x-coordinate of the end cross section point
36-45	F10.0	END(2)	y-coordinate of the end cross section point
46-50	I5	NX	No. of x-values input
51-55	I5	NY	No. of y-values input
56-60	I5	NUMPTS	No. of points to be used for calculating the cross section
61-65	I5	SETS	No. of data sets to be plotted
66-70	I5	NPEN	No. of different colored pens to use (choice of 1,2 or 3 colors to alternate when drawing cross section)
Card 2			
1-10	F10.0	XUPIN	Units/inch in the x-direction
11-20	F10.0	YINCH	If > 0 No. of inches in the y-direction If ≤ 0 use XUPIN as scale for y-direction
21-40	5A4	XTITLE	Label written on x-axis
41-60	5A4	YTITLE	Label written on y-axis
Card 3			
1-30	10I3	SET(I)	List (by order in which they are read in) of each data set that will be used in drawing the cross-section (up to 10)
Card 4			
1-80	20A4	TITLE	Heading to appear at the top of the plot

COLUMNS	FORMAT	NAME	DESCRIPTION
NX/8 Cards			
1-80	8F10.0	X(J)	x-coordinates of grid points
NY/8 Cards			
1-80	8F10.0	Y(K)	y-coordinates of grid points
*ZS*NX*NY/8 Cards*			
1-80	8F10.0	Z(I,J,K)	Z-VALUES to be plotted I = No. Sets J = No. x's K = No. y's

EXAMPLE

This example demonstrates the use of the program in constructing geologic cross-sections from structural maps. Figure 5 shows a finite-difference grid and the location of the section A-A'. Four sets of z-values were used in this example, and the input is shown in Table 1. Each data set is composed of elevation data on each of four geologic units; a structural map at the top of "Unit 3" (see figure 7) is shown in figure 6. The elevations in figure 4 are in meters above mean sea level and were prepared for a finite-difference model using well data. The contours were plotted using the Calcomp GPCP program. Using the structural map in figure 6 and the other elevation data, the geologic cross-section A-A' in figure 7 was constructed.



Figure 5. Areal finite-difference grid showing location of cross-section A-A'. Distance is in meters.

4 3048.00 7315.00 10667.00 2743.20 16 10 40 4 1

609.6 -1.0 METERS METERS

1 2 3 4

NORTHWEST TO SOUTHEAST CROSS SECTION

609.60	1828.80	3048.00	3962.40	4572.00	5181.60	5791.20	6248.4
6553.20	6858.00	7315.20	7772.40	8077.20	8534.40	9448.80	10668.0
609.60	1828.80	2743.20	3352.80	3962.40	4419.60	4724.40	5181.6
6096.00	7315.00						
-0.0	0.0	-635.00	-645.00	-655.00	-710.00	-745.00	-895.0
0.0	0.0	0.0	-635.00	-640.00	-645.00	-680.00	-725.0
-755.00	-755.00	-780.00	0.0	-635.00	-640.00	-645.00	-645.0
-690.00	-725.00	-745.00	-760.00	-700.00	-510.00	-640.00	-645.0
-645.00	-650.00	-670.00	-700.00	-730.00	-695.00	-545.00	-345.0
-640.00	-645.00	-545.00	-515.00	-590.00	-680.00	-650.00	-550.0
-360.00	-355.00	-645.00	-645.00	-490.00	-340.00	-275.00	-425.0
-460.00	-360.00	-265.00	-305.00	-645.00	-645.00	-500.00	-270.0
-125.00	-220.00	-215.00	-240.00	-230.00	-265.00	-650.00	-655.0
-565.00	-325.00	-180.00	-230.00	-265.00	-255.00	-200.00	-215.0
-650.00	-655.00	-625.00	-450.00	-235.00	-200.00	-220.00	-220.0
-175.00	-165.00	-655.00	-695.00	-685.00	-585.00	-315.00	-190.0
-180.00	-180.00	-145.00	-130.00	-695.00	-755.00	-800.00	-700.0
-505.00	-350.00	-300.00	-195.00	-65.00	-70.00	-755.00	-820.0
-905.00	-905.00	-735.00	-665.00	-550.00	-350.00	-75.00	-65.0
-805.00	-870.00	-980.00	-1005.00	-955.00	-860.00	-745.00	-550.0
-165.00	-60.00	-880.00	-970.00	-1035.00	-1035.00	-995.00	-930.0
-880.00	-775.00	-400.00	-70.00	0.0	-1040.00	-1040.00	-1040.0
-1035.00	-1005.00	-950.00	-845.00	-705.00	-320.00	0.0	0.0
-1040.00	-1040.00	-1040.00	-1035.00	-1010.00	-910.00	-710.00	0.0
0.0	0.0	395.00	385.00	375.00	320.00	285.00	135.0
0.0	0.0	0.0	395.00	390.00	385.00	350.00	305.0
270.00	220.00	100.00	0.0	395.00	390.00	385.00	385.0
340.00	275.00	220.00	150.00	75.00	40.00	390.00	385.0
380.00	375.00	330.00	235.00	170.00	135.00	80.00	45.0
350.00	385.00	365.00	360.00	320.00	220.00	175.00	150.0
90.00	35.00	385.00	380.00	375.00	345.00	350.00	250.0
175.00	160.00	125.00	85.00	385.00	380.00	375.00	350.0
325.00	260.00	250.00	180.00	160.00	125.00	380.00	375.0
355.00	325.00	270.00	250.00	260.00	255.00	210.00	175.0
380.00	375.00	325.00	300.00	265.00	250.00	260.00	260.0
250.00	225.00	375.00	335.00	290.00	265.00	260.00	260.0
260.00	260.00	285.00	260.00	335.00	275.00	225.00	200.0
195.00	200.00	200.00	225.00	325.00	320.00	275.00	210.0
125.00	95.00	115.00	135.00	150.00	200.00	325.00	325.0
225.00	160.00	50.00	20.00	70.00	115.00	130.00	200.0
325.00	330.00	150.00	60.00	-5.00	-5.00	35.00	100.0
150.00	225.00	325.00	330.00	0.0	-10.00	-10.00	-10.0
-5.00	25.00	80.00	185.00	325.00	330.00	0.0	0.0
-10.00	-10.00	-10.00	-5.00	20.00	120.00	320.00	0.0
0.0	0.0	430.00	420.00	415.00	385.00	365.00	230.0
0.0	0.0	0.0	430.00	425.00	420.00	390.00	395.0
380.00	340.00	235.00	0.0	430.00	425.00	420.00	420.0
380.00	350.00	345.00	305.00	235.00	200.00	425.00	420.0
415.00	410.00	380.00	350.00	340.00	325.00	245.00	205.0
425.00	420.00	400.00	395.00	395.00	385.00	385.00	375.0
255.00	195.00	420.00	415.00	410.00	380.00	405.00	380.0
385.00	380.00	295.00	245.00	420.00	415.00	410.00	385.0
375.00	335.00	350.00	340.00	340.00	285.00	415.00	415.0

Table 1. Listing of input data for example problem.

355.00	375.00	335.00	320.00	325.00	330.00	380.00	330.
415.00	425.00	390.00	365.00	330.00	320.00	310.00	315.
375.00	360.00	415.00	405.00	365.00	345.00	340.00	320.
315.00	305.00	335.00	310.00	410.00	380.00	340.00	320.
330.00	335.00	320.00	275.00	365.00	365.00	390.00	355.
320.00	315.00	330.00	315.00	300.00	270.00	365.00	365.
365.00	340.00	325.00	345.00	330.00	295.00	265.00	280.
365.00	370.00	355.00	330.00	325.00	325.00	320.00	310.
290.00	315.00	370.00	370.00	0.0	325.00	325.00	325.
325.00	325.00	305.00	320.00	380.00	370.00	0.0	0.
330.00	330.00	325.00	325.00	330.00	330.00	380.00	0.
0.0	0.0	640.00	630.00	625.00	595.00	575.00	440.
0.0	0.0	0.0	620.00	605.00	595.00	555.00	545.
535.00	515.00	455.00	0.0	600.00	585.00	570.00	555.
515.00	515.00	525.00	510.00	445.00	420.00	575.00	550.
530.00	530.00	515.00	510.00	525.00	530.00	445.00	380.
560.00	540.00	505.00	510.00	530.00	535.00	550.00	555.
425.00	340.00	545.00	525.00	485.00	480.00	530.00	520.
515.00	530.00	435.00	360.00	515.00	495.00	475.00	470.
475.00	430.00	465.00	475.00	460.00	370.00	480.00	465.
450.00	445.00	435.00	440.00	455.00	465.00	490.00	405.
460.00	470.00	440.00	450.00	435.00	420.00	435.00	445.
475.00	425.00	445.00	445.00	410.00	445.00	440.00	390.
415.00	430.00	425.00	360.00	435.00	415.00	395.00	415.
395.00	410.00	420.00	380.00	445.00	390.00	410.00	380.
370.00	390.00	400.00	380.00	390.00	355.00	425.00	385.
395.00	360.00	360.00	400.00	390.00	380.00	345.00	360.
405.00	350.00	375.00	350.00	345.00	360.00	370.00	375.
355.00	365.00	395.00	390.00	0.0	345.00	345.00	345.
345.00	345.00	325.00	340.00	400.00	390.00	0.0	0.
345.00	345.00	340.00	340.00	345.00	345.00	395.00	0.

Table 1. Cont.



Figure 6. Structural map at the top of "Unit 3". Contour interval is 25 meters.



Figure 7. Plot of geologic cross-section A-A' obtained from CSPP.

```
C   *****************************************************************
C   *
C   *      PROGRAM LISTING
C   *
C   *****************************************************************
      REAL BEG(2),END(2),PX(100),PXX(102),PY(100),PYY(102),TITLE(20)
      REAL X(50),Y(50),Z(10,50,50),ZZ(50,50),XTITLE(5),YTITLE(5)
      INTEGER SET(10),SETS,XYZS,YMAX,ZS,IBUF(1000)
C
C
C BEG-    X AND Y COORDINATES OF THE FIRST CROSS SECTION POINT
C END-    X AND Y COORDINATES OF THE LAST CROSS SECTION POINT
C IBUF-  USED IN 'PLOTS' ROUTINE
C  NUMPTS-# OF POINTS GENERATED FOR CROSS SECTION
C  NX-      NUMBER OF X-VALUES
C  NY-      NUMBER OF Y-VALUES
C  UPI-    UNITS/INCH IN Y-DIRECTION
C PX-     X COORDINATE OF THE CALCULATED CROSS SECTION POINTS
C PXX-    USED IN DRAWING ROUTINE AS LOCATIONS ON THE X-AXIS
C PY-     Y COORDINATES AS IN PX
C PYY-    CALCULATED Z VALUES IN DRAWING ROUTINE
C SET-    THE ORDER OF THE PLOTS TO BE DRAWN
C SETS-  THE # OF PLOTS TO BE DRAWN
C TITLE- AT TOP OF PLOT
C X,Y,Z- GRID SYSTEM READ IN AT THE BEGINING
C  XUPIN- UNITS/INCH IN X-DIRECTION
C  YINCH- # INCHES IN Y-DIRECTION
C  XTITLE-LABEL FOR X-AXIS
C  YTITLE-LABEL FOR Y-AXIS
C ZS-     THE # OF SETS OF Z'S AVAILABLE
C  ZZ-     TEMPORARY STORAGE FOR Z-VALUES
C
C
      CALL PLOTS(IBUF,1000,9)
C
C  INPUT
C     1) #OF SETS OF Z'S TO READ
C     2) X-COORDS FOR FIRST CROSS SECTION POINT
C     3) Y-COORDS FOR FIRST CROSS SECTION POINT
C     4) X-COORDS FOR LAST CROSS SECTION POINT
C     5) Y-COORDS FOR LAST CROSS SECTION POINT
C     6) #OF GRID POINTS
C     7) # OF CROSS SECTION POINTS DESIRED
C     8) # OF SETS TO BE PLOTED
C     9) # OF PENS TO USE FOR THE LINES
C    10) UNITS/INCH IN THE X-DIRECTION
C    11) INCHES IN THE Y-DIRECTION
C     12) LABEL FOR X-AXIS
C     13) LABEL FOR Y-AXIS
C     14) WHICH 'Z' SETS TO PLOT
C     15) TITLE
C     16) X,Y,Z1,Z2,Z3,...
C
      READ (5,90) ZS,BEG,END,NX,NY,NUMPTS,SETS,NPEN
      READ (5,100) XUPIN,YINCH,XTITLE,YTITLE
      READ (5,110) (SET(I),I=1,SETS)
      READ (5,120) TITLE
      READ (5,130) (X(I),I=1,NX)
      READ (5,130) (Y(I),I=1,NY)
      IF (BEG(1).LT.X(1).OR.BEG(1).GT.X(NX)) GO TO 70
```

```
      IF (BEG(2).LT.Y(1).OR.BEG(2).GT.Y(NY)) GO TO 70
      IF (END(1).LT.X(1).OR.END(1).GT.X(NX)) GO TO 70
      IF (END(2).LT.Y(1).OR.END(2).GT.Y(NY)) GO TO 70
      DO 10 K=1,ZS
   10 READ (5,130) ((Z(K,I,J),J=1,NY),I=1,NX)
C
C  AND FIND LABELS FOR FOR VALUES TO BE WRITTEN ON THE X-AXIS
C  FIND X & Y COORDINATES OF NEWLY GENERATED CROSS SECTION POINTS
C
      NUMPTS=NUMPTS-1
      XLEN=(BEG(1)-END(1))/FLOAT(NUMPTS)
      YLEN=(BEG(2)-END(2))/FLOAT(NUMPTS)
      CRSLEN=SQRT(ABS(BEG(1)-END(1))**2+ABS(BEG(2)-END(2))**2)
      PX(1)=BEG(1)
      PY(1)=BEG(2)
      PXX(1)=0.0
      DO 20 I=2,NUMPTS
      PY(I)=PY(I-1)-YLEN
      PXX(I)=PXX(I-1)+CRSLEN/FLOAT(NUMPTS)
   20 PX(I)=PX(I-1)-XLEN
      NUMPTS=NUMPTS+1
      PX(NUMPTS)=END(1)
      PY(NUMPTS)=END(2)
      PXX(NUMPTS)=PXX(NUMPTS-1)+CRSLEN/FLOAT(NUMPTS-1)
      XINCH=CRSLEN/XUPIN
C
C  DETERMINE AND ASSIGN SCALING FACTORS FOR PLOT SUBROUTINES
C
      PXX(NUMPTS+1)=PXX(1)
      PXX(NUMPTS+2)=XUPIN
      ZMAX=Z(1,1,1)
      ZMIN=Z(1,1,1)
      DO 30 K=1,SETS
      DO 30 I=1,NX
      DO 30 J=1,NY
      IF (Z(K,I,J).LT.ZMIN) ZMIN=Z(K,I,J)
   30 IF (Z(K,I,J).GT.ZMAX) ZMAX=Z(K,I,J)
      PYY(NUMPTS+1)=ZMIN
      UPI=(ZMAX-ZMIN)/YINCH
      IF (YINCH.LE.0.) UPI=XUPIN
      IF (YINCH.LE.0.) YINCH=(ZMAX-ZMIN)/UPI
      PYY(NUMPTS+2)=UPI
      PXAR=XUPIN/UPI
C
C  BEGIN LOOP FOR CONSECUTIVE 'Z'SETS
C  BEGIN LOOP TO FIND VALUES AT CROSS SECTION POINTS
C
      M=1
      DO 60 J=1,SETS
      DO 40 K=1,NX
      DO 40 I=1,NY
      ZZ(K,I)=Z(SET(J),K,I)
   40 CONTINUE
      WRITE (6,140) SET(J)
      DO 50 I=1,NUMPTS
      CALL GETVAL(PX(I),PY(I),X,Y,ZZ,NX,NY,VALUE)
   50 PYY(I)=VALUE
```

```
      WRITE (6,150) (I,PX(I),PY(I),PYY(I),I=1,NUMPTS)
      IF (N.GT.NPEN) N=1
      CALL NEWPEN(N)
      N=N+1
   60 CALL LINE(PXX,PYY,NUMPTS,1,0,0)
C
C  DRAW BOUNDARIES AND LABELS
C
      CALL NEWPEN(1)
      TEMP=PXX(NUMPTS+1)
      CALL AXIS(0.0,0.0,XTITLE,-20,XINCH,0.0,TEMP,PXX(NUMPTS+2))
      TEMP=PYY(NUMPTS+1)
      CALL AXIS(0.0,0.0,YTITLE,20,YINCH,90.0,TEMP,PYY(NUMPTS+2))
      CALL PLOT(XINCH,0.0,2)
      CALL PLOT(XINCH,YINCH,2)
      CALL SYMBOL(.5,YINCH+1.0,.28,TITLE,0.,80)
      CALL SYMBOL(.5,YINCH+.5,.14,'VERTICAL EXAGERATION = ',0.,22)
      CALL NUMBER(999.,999.,.14,EXAG,0.,2)
      CALL SYMBOL(999.,999.,.14,' X-UNITS/Y-UNITS',0.,16)
      CALL SYMBOL(0.,-.56,.14,'( ',0.,2)
      CALL NUMBER(999.,999.,.14,BEG(1),0.,2)
      CALL SYMBOL(999.,999.,.14,' , ',0.,3)
      CALL NUMBER(999.,999.,.14,BEG(2),0.,2)
      CALL SYMBOL(999.,999.,.14,' )',0.,2)
      CALL SYMBOL(XINCH-2.38,-.56,.14,'( ',0.,2)
      CALL NUMBER(999.,999.,.14,END(1),0.,2)
      CALL SYMBOL(999.,999.,.14,' , ',0.,3)
      CALL NUMBER(999.,999.,.14,END(2),0.,2)
      CALL SYMBOL(999.,999.,.14,' )',0.,2)
      CALL PLOT(0.,YINCH+2.5,999)
      GO TO 80
   70 WRITE (6,160) BEG,END
   80 STOP
C
C
   90 FORMAT (I2,3X,4F10.2,5I5)
  100 FORMAT (2F10.0,2(5A4))
  110 FORMAT (10I3)
  120 FORMAT (20A4)
  130 FORMAT (8F10.2)
  140 FORMAT (' SET #',I5)
  150 FORMAT (I10,'   X =',F10.2,'   Y =',F10.2,'   VALUE =',F10.2)
  160 FORMAT (' CROSS SECTION END POINTS SPECIFIED BY THE USER ARE ','OU
     1TSIDE THE BOUNDARIES OF THE GRID:'/' BEG. POINT = (',F10.2,',',F10
     2.2,' )   END POINT = (',F10.2,',',F10.2,' )')
      END
```

```
C  THIS SUBROUTINE CALCULATES BY INTERPOLATION THE Z VALUES AT THE CROSS
C  SECTION POINT USING THE FOUR GRID POINTS FORMING A BOX AROUND THE POI
C XP-     X-COORDINATE OF POINT AT WHICH VALUE IS TO BE CALCULATED
C YP-     Y-COORDINATE OF POINT AS WITH XP
C X-      50 ELEMENT ARRAY OF X-COORDINATES
C Y-      50 ELEMENT ARRAY OF Y-COORDINATES
C Z-      50X50 ELEMENT ARRAY OF Z-VALUES
C NX-     # OF X-COORDINATES ACTUALLY USED IN THE ARRAY
C NY-     # OF Y-COORDINATES ACTUALLY USED IN THE ARRAY
C VALUE-THE Z-VALUE OF POINT THAT IS RETURNED TO CALLING ROUTINE
C
C
      SUBROUTINE GETVAL(XP,YP,X,Y,Z,NX,NY,VALUE)
      DIMENSION X(50), Y(50), Z(50,50)
      ETA(A,A1,A2)=(2.*A-(A1+A2))/(A2-A1)
C
C  FIND GRID BOX SURROUNDING POINT.
C  IF POINT IS NOT IN THE GRID THEN END THE PROGRAM
C
      DO 10 I=1,NX
      IF (XP.GT.X(I)) GO TO 10
      II=I
      GO TO 20
   10 CONTINUE
   20 DO 30 J=1,NY
      IF (YP.GT.Y(J)) GO TO 30
      JJ=J
      GO TO 40
   30 CONTINUE
C
C  CALCULATE Z-VALUES FROM Z-VALUES OF THE FOUR POINTS FOUND ABOVE.
C
   40 Z1=Z(II-1,JJ-1)
      Z2=Z(II,JJ-1)
      Z3=Z(II,JJ)
      Z4=Z(II-1,JJ)
      AN=ETA(YP,Y(JJ-1),Y(JJ))
      E=ETA(XP,X(II-1),X(II))
      V1=.25*(1.-AN)*(1.-E)
      V2=.25*(1.-AN)*(1.+E)
      V3=.25*(1.+AN)*(1.+E)
      V4=.25*(1.+AN)*(1.-E)
      VALUE=V1*Z1+V2*Z2+V3*Z3+V4*Z4
      RETURN
      END
```

REFERENCES

California Computer Products, Inc., 1971, GPCP A General Purpose Contouring Program: User's Manual.

California Computer Products, Inc., 1969, Programming Calcomp Pen Plotters.

Zienkiewicz, O. C., 1971, The Finite Element Method in Engineering Science, London, McGraw-Hill Book Co., Inc., 521 p.